
Intel's IA32 instruction set architecture (ISA), colloally known as “x86”, is the dominant instruction for-
mat for the world’s computers. 1A32 is the platform of chofoe most Windows, Linux, and, since 2006,
even Macintosh computers. The ISA we use today was define@88 With the introduction of the i386
microprocessor, extending the 16-bit instruction set eefiny the original 8086 to 32 bits. Even though
subsequent processor generations have introduced newciitsh types and formats, many compilers, in-
cludinggcc, have avoided using these features in the interest of naimggbackward compatibility.

A shift is underway to a 64-bit version of the Intel instractiset. Originally developed by Advanced Micro
Devices (AMD) and namegl86-64 it is now supported by high end processors from AMD (who nedl ¢
it AMDG64) and by Intel, who refer to it amtel64. Most people still refer to it as “x86-64,” and we follow
this convention. (Some vendors have shortened this to gifwpl”). Newer versions of Linux andcc
support this extension. In making this switch, the devealsm Gcc saw an opportunity to also make use
of some of the instruction-set features that had been addedie recent generations of IA32 processors.

This combination of new hardware and revised compiler makés64 code substantially different in form
and in performance than IA32 code. In creating the 64-bmrsion, the AMD engineers also adopted some
of the features found in reduced-instruction set compyRISC) [7] that made them the favored targets for
optimizing compilers. For example, there are now 16 geraugbose registers, rather than the performance-
limiting eight of the original 8086. The developers ®tc were able to exploit these features, as well as
those of more recent generations of the IA32 architectorebtain substantial performance improvements.
For example, procedure parameters are now passed viaersgiather than on the stack, greatly reducing
the number of memory read and write operations.

This document serves as a supplement to Chapter@oafputer Systems: A Programmer’'s Perspective
(CS:APP), describing some of the differences. We start aithief history of how AMD and Intel arrived

at x86-64, followed by a summary of the main features thdtrdjeish x86-64 code from 1A32 code, and
then work our way through the individual features.

*Copyright(© 2005, 2008, R. E. Bryant, D. R. O’Hallaron. All rights resetv



1 History and Motivation for x86-64

Over the more than twenty years since introduction of thé&,i3Be capabilities of microprocessors have
changed dramatically. In 1985, a fully configured, high-eledktop computer had around 1 megabyte
of random-access memory (RAM) and 50 megabytes of disk ggoravlicroprocessor-based “worksta-
tion” systems were just becoming the machines of choice donputing and engineering professionals.
A typical microprocessor had a 5-megahertz clock and raorarane million instructions per second.
Nowadays, a typical high-end system has 2 gigabyte of RANK(#@rease), 1 terabyte of disk storage
(20,000X increase), and a 4-gigahertz clock, running atdabillion instructions per second (5000X in-
crease). Microprocessor-based systems have become ipervasen today’s supercomputers are based
on harnessing the power of many microprocessors computiqrnallel. Given these large quantitative
improvements, it is remarkable that the world’s computiagdomostly runs code that is binary compatible
with machines that existed over 20 years ago.

The 32-bit word size of the IA32 has become a major limitatiogrowing the capacity of microprocessors.
Most significantly, the word size of a machine defines the eavfgrirtual addresses that programs can use,
giving a 4-gigabyte virtual address space in the case of &2 It is now feasible to buy more than this
amount of RAM for a machine, but the system cannot make @feease of it. For applications that involve
manipulating large data sets, such as scientific computi@igibases, and data mining, the 32-bit word size
makes life difficult for programmers. They must write codengut-of-corealgorithms, where the data
reside on disk and are explicitly read into memory for preges

Further progress in computing technology requires a shiét ftarger word size. Following the tradition of
growing word sizes by doubling, the next logical step is 64.dn fact, 64-bit machines have been available
for some time. Digital Equipment Corporation introducesl Alpha processor in 1992, and it became
a popular choice for high-end computing. Sun Microsystemiduced a 64-bit version of its SPARC
architecture in 1995. At the time, however, Intel was notrégosis contender for high-end computers, and
so the company was under less pressure to switch to 64 bits.

Intel's first foray into 64-bit computers were the Itaniunopessors, based on the 1A64 instruction set.
Unlike Intel’s historic strategy of maintaining backwarnapatibility as it introduced each new generation
of microprocessor, 1A64 is based on a radically new apprqgaitily developed with Hewlett-Packard.
Its Very Large Instruction Word@VLIW) format packs multiple instructions into bundles|aabing higher
degrees of parallel execution. Implementing IA64 provebdeaery difficult, and so the first Itanium chips
did not appear until 2001, and these did not achieve the égdevel of performance on real applications.
Although the performance of Itanium-based systems hasowepl; they have not captured a significant
share of the computer market. Itanium machines can exef32 ¢ode in a compatibility mode but not
with very good performance. Most users have preferred toendakwith less expensive, and often faster,
IA32-based systems.

Meanwhile, Intel’s archrival, Advanced Micro Devices (ANIBaw an opportunity to exploit Intel’s misstep
with 1A64. For years AMD had lagged just behind Intel in teology, and so they were relegated to
competing with Intel on the basis of price. Typically, Intebuld introduce a new microprocessor at a
price premium. AMD would come along 6 to 12 months later angehta undercut Intel significantly to

1The physical memory of a machine is often referred te@® memorydating to an era when each bit of a random-access
memory was implemented with a magnetized ferrite core.



get any sales—a strategy that worked but yielded very loviiteroln 2002, AMD introduced a 64-bit
microprocessor based on its “x86-64" instruction set. Asrlhme implies, x86-64 is an evolution of the
Intel instruction set to 64 bits. It maintains full backwazdmpatibility with 1A32, but it adds new data
formats, as well as other features that enable higher dgpawil higher performance. With x86-64, AMD
has sought to capture some of the high-end market that heatibély belonged to Intel. AMD’s recent
generations of Opteron and Athlon 64 processors have ingiee@d very successful as high performance
machines. Most recently, AMD has renamed this instructienAdMD64, but “x86-64" persists as the
favored name.

Intel realized that its strategy of a complete shift from 2A® 1A64 was not working, and so began sup-
porting their own variant of x86-64 in 2004 with processardhie Pentium 4 Xeon line. Since they had
already used the name “IA64” to refer to Itanium, they thesetha difficulty in finding their own name for
this 64-bit extension. In the end, they decided to descr&64 as an enhancement to 1A32, and so they
referred to it adA32-EM64Tfor “Enhanced Memory 64-bit Technology.” In late 2006 thelopted the
namelntel64.

On the compiler side, the developersagfc steadfastly maintained binary compatibility with the i388en

as useful features were being added to the IA32 instructgin $he 1995 PentiumPro introduced a set
of conditional move instructions that could greatly impedihe performance of code involving conditional
operations. More recent generations of Pentium processwosluced new floating point operations that
could replace the rather awkward and quirky approach datieg to the 8087, the floating point coprocessor
that accompanied the 8086 and, since the i486, has beerdéucks part of the microprocessor itseulf.
Switching to x86-64 as a target provided an opportunitydarc to give up backward compatibility and
instead exploit these newer features.

In this document, we use “IA32" to refer to the combinatiorhafdware andcc code found in traditional,
32-bit versions of Linux running on Intel-based machinese We “x86-64" to refer to the hardware and
code combination running on the newer 64-bit machines frdtDAand Intel. In the Linux world, these
two platforms are referred to as “i386” and “x&@l,” respectively.

2 Finding Documentation

Both Intel and AMD provide extensive documentation on tipedcessors. This includes general overviews
of the assembly language programmer’s view of the hardward][ as well as detailed references about
the individual instructions [3, 5, 6]. The organizatiamd64.org has been responsible for defining the
Application Binary Interfac€ABI) for x86-64 code running on Linux systems [8]. This irfeece describes
details for procedure linkages, binary code files, and a murabother features that are required for object
code programs to execute properly.

Warning: Both the Intel and the AMD documentation use thellassembly code notation. This differs
from the notation used by the Gnu assemlias. Most significantly, it lists operands in the opposite order



C declaration Intel data type GAS suffix | x86-64 Size (Bytes
char Byte b 1
short Word w 2
int Double word I 4
unsigned Double word I 4
long int Quad word q 8
unsigned long Quad word q 8
char = Quad word q 8
float Single precision S 4
double Double precision d 8
long double Extended precisior t 16

Figure 1:Sizes of standard data types with x86-64 Both long integers and pointers require 8 bytes, as
compared to 4 for IA32.

3 An Overview of x86-64

The combination of the new hardware supplied by Intel and AMB well as the new version @cc
targeting these machines makes x86-64 code substantitiflyetit from that generated for IA32 machines.
The main features include:

e Pointers and long integers are 64 bits long. Integer ariticno@erations support 8, 16, 32, and 64-bit
data types.

e The set of general-purpose registers is expanded from 8.to 16

e Much of the program state is held in registers rather tharherstack. Integer and pointer procedure
arguments (up to 6) are passed via registers. Some proseadluret need to access the stack at all.

e Conditional operations are implemented using conditiomaVve instructions when possible, yielding
better performance than traditional branching code.

e Floating-point operations are implemented using a reg@iented instruction set, rather than the
stack-based approach supported by IA32.

3.1 Data Types

Figure 1 shows the sizes of different C data types for x86€&@Imparing these to the IA32 sizes (CS:APP
Figure 3.1), we see that pointers (shown here as datadgpe *) require 8 bytes rather than 4. In
principal, this gives programs the ability to accesst@bytef memory (around 8.4 x 10'® bytes). That
seems like an astonishing amount of memory, but keep in nhiad4t gigabytes seemed astonishing when
the first 32-bit machines appeared in the late 1970s. Inipeaanost machines do not really support the
full address range—the current generations of AMD and k8664 machines support 256 terabyt2®’)
bytes of virtual memory—but allocating a full 64 bits for pters is a good idea for long term compatibility.



We also see that the prefixdhg ” changes integers to 64 bits, allowing a considerably largage of
values. Whereas a 32-bit unsigned value can range up t0,9&8295 (CS:APP Figure 2.8), increasing
the word size to 64 bits gives a maximum value of 18,446,743} 109,551,615.

As with 1A32, thelong prefix also changes a floating poitbuble to use the 80-bit format supported

by I1A32 (CS:APP Section 2.4.6.) These are stored in memotly an allocation of 16 bytes for x86-64,

compared to 12 bytes for IA32. This improves the performaieeemory read and write operations, which
typically fetch 8 or 16 bytes at a time. Whether 12 or 16 bytesatiocated, only the low-order 10 bytes are
actually used.

3.2 Assembly Code Example

Section 3.2.3 of CS:APP illustrated the 1A32 assembly cosleegated byccc for a functionsimple .

Below is the C code fosimple 1, similar tosimple , except that it uses long integers:
long int simple_l(long int *Xp, long int y)
{
long int t = *Xp + Y,
*Xp =
return t;
}

Whenaccis run on an x86-64 machine with the command line
unix> gcc -2 -S -nB2 code.c
it generates code that is compatible with any IA32 machine:

| A32 version of function sinple_l.
Argunments in stack | ocations 8(%bp) (xp) and 12(%bp) (y)

1 simple_l:

2 pushl %ebp Save frane pointer

3 movl %esp, %ebp Create new frame pointer

4 movl 8(%ebp), Yedx Get xp

5 movl (%edx), Yeax Retrieve *xp

6 addl 12(%ebp), %eax Add y to get t (and return val ue)
7 movl %eax, (Yoedx) Store t at *xp

8 leave Restore stack and frane pointers
9 ret Return

This code is almost identical to that shown in CS:APP, extlegt it uses the singleeave instruction
(CS:APP Section 3.7.2), rather than the sequenoel %ebp, %esp andpopl %ebp to deallocate the
stack frame.

When we instructsccto generate x86-64 code
unix> gcc -2 -S -nbB4 code.c

(on most machines, the flag64 is not required), we get very different code:

x86-64 version of function sinple_|.
Argunments in registers %di (xp) and %si (y)

1 simple_l:

2 addq (%rdi), %orsi Add *xp toy to get t
3 movq %0rsi, %rax Set t as return val ue
4 movq %rsi, (%ordi) Store t at *xp

5 ret Return

Some of the key differences include

Instead oimovl andaddl instructions, we semovqgandaddq . The pointers and variables declared
as long integers are now 64 bits (quad words) rather thant8Zlbng words).

We see the 64-bit versions of the registers, €grsi , %rdi . The procedure returns a value by
storing it in registeforax.

No stack frame gets generated in the x86-64 version. Thigraites the instructions that set up (lines
2-3) and remove (line 8) the stack frame in the 1A32 code.

Argumentsxp andy are passed in registe¥srdi and%rsi , rather than on the stack. These registers
are the 64-bit versions of registeé¥sedi and%esi . This eliminates the need to fetch the arguments
from memory. As a consequence, the two instructions on Enasd 3 can retrievexp, add it toy,
and set it as the return value, whereas the IA32 code reqthired lines of code: 4-6.

The net effect of these changes is that the 1A32 code condisftsinstructions making 7 memory refer-
ences, while the x86-64 code consists of 4 instructions mgaBimemory references. Running on an Intel
Pentium 4 Xeon, our experiments show that the IA32 code reg@round 17 clock cycles per call, while
the x86-64 code requires 12 cycles per call. Running on an Abfiteron, we get 9 and 7 cycles per call,
respectively. Getting a performance increase of 1.3—14Xhe same machine with the same C code is a
significant achievement. Clearly x86-64 represents an itapbstep forward.

4 Accessing Information

Figure 2 shows the set of general-purpose registers unde648 Compared to the registers for IA32
(CS:APP Figure 3.2), we see a number of differences:

The number of registers has been doubled to 16. The neweegate numbered 8-15.

All registers are 64 bits long. The 64-bit extensions of tA82 registers are namédrax, %orcx,
%rdx, %rbx, %rsi , %rdi , %rsp, and%rbp. The new registers are namea8—%r15.

The low-order 32 bits of each register can be accessed lgliréidtis gives us the familiar registers
from 1A32: %eax, %ecx, %oedx, %ebx, %esi, %edi, %esp, and%ebp, as well as eight new 32-bit
registers:%r8d—%r15d.

The low-order 16 bits of each register can be accessed lgirastis the case for IA32. Theord-size
versions of the new registers are nanded@w%r15w.

63 31

[

5 87 0

%rax Y%eax %ax| %ah | %al Return value

%rbx %ebx %Dbx| %bh | %bl Callee saved

%rcx %ecx %cx| %ch | %cl | 4th argument

%rdx %edx %dx| %dh | %dl | 3rd argument

%rsi %esi %si %sil 2nd argument

%rdi %edi %di %dil 1st argument

%rbp %ebp  %bp %bpl | Callee saved

%rsp %esp  %sp %spl || Stack pointer

%r8 %r8d  %r8w, %r8b || 5th argument

%r9 %r9d  %r9w, %r9b || 6th argument

%r10 %r10d %rl10w %rl0b || Callee saved

%r11 %rlld %rllw %r11b || Used for linking

%r12 %rl12d %rl2w %r12b | Unused for C

%rl13 %r13d %rl13w %r13b || Callee saved

%r14 %r14d %rldw %rl4b | callee saved

%rl15 %r15d %rl5w %r15b | Callee saved

Figure 2:Integer registers. The existing eight registers are extended to 64-bit versions, and eight new
registers are added. Each register can be accessed as either 8 bits (byte), 16 bits (word), 32 bits (double
word), or 64 bits (quad word).



e The low-order 8 bits of each register can be accessed dirddtis is true in IA32 only for the first 4
registers oal , %cl , %dl, %bl). The byte-size versions of the other IA32 registers areat¥sil ,
%dil , %spl, and%bpl. The byte-size versions of the new registers are navhib—%r15b.

e For backward compatibility, the second byte of registnsax, %rcx, %rdx, and%rbx can be
directly accessed by instructions having single-byte apes.

As with IA32, most of the registers can be used intercharngehbt there are some special cases. Register
%rsp has special status, in that it holds a pointer to the top stéakent. Unlike in IA32, however, there
is no frame pointer register; registeétrbp is available for use as a general-purpose register. Plarticu
conventions are used for passing procedure argumentsgisars and for how registers are to be saved and
restored during procedure calls, as is discussed in Segtibmaddition, some arithmetic instructions make
special use of registefgrax and%rdx.

For the most part, the operand specifiers of x86-64 are jessdme as those in IA32 (see CS:APP Fig-
ure 3.3). One minor difference is that some form$>&f-relativeoperand addressing are supported. With
IA32, this form of addressing is only supported for jump attteo control transfer instructions (see CS:APP
Section 3.6.3). This mode is provided to compensate fordhethat the offsets (shown in CS:APP Fig-
ure 3.3 agmm) are only 32 bits long. By viewing this field as a 32-bit, two@mplement number, instruc-
tions can access data within a window of arodfi15 x 10? relative to the program counter. With x86-64,
the program counter is namétkrip .

As an example of PC-relative data addressing, considerolfeving procedure, which calls the function
call _simple 1 examined earlier:

567,
763;

long int gvall
long int gval2

long int call_simple_I()

{
long int z = simple_I(&gvall, 12L);
return z + gval2;

}

This code references global variabtpsall andgval2 . When this function is compiled, assembled, and
linked, we get the following executable code (as generayettid disassemblaybjdump )

1 0000000000400500 <call_simple_I>:

2 400500: be Oc 00 00 00 mov $0xc,%esi Load 12 as 1st argunent

3 400505: bf 08 12 50 00 mov $0x501208,%edi Load &gval 1 as 2nd ar gunment
4  40050a: e8 bl ff ff ff callg 4004cO0 <simple I> Call sinple_|

5  40050f: 48 03 05 ea Oc 10 00 add 1051882 (%orip),%rax Add gval 2 to result

6 400516: c3 retq

The instruction on line 3 stores the address of global veigdall inregisteordi . It does this by simply
copying the constant valuex501208 into register%edi. The upper 32 bits ddordi are then automat-
ically set to zero. The instruction on line 5 retrieves thkigaofgval2 and adds it to the value returned



Instruction Effect Description

mov(q S,D|D «— S Move quad word

movabsq I,R | R « I Move absolute quad word
movslq S,R | R «— SignExtend(S5) Move sign-extended double word
movsbg S, R | R <« SignExtend(S) Move sign-extended byte
movzbq S,R | R <« ZeroExtend(S) Move zero-extended byte

pushq S R[%rsp] <« R[%rsp] —8; | Push

M[R[%rsp]] <« S
popq D D — M[R[%rsp][; Pop
R[%rsp] «— R[%rsp]+ 8

Figure 3. Data movement instructions. These supplement the movement instructions of 1A32. The
movabsq instruction only allows immediate data (shown as I) as the source value. Others allow, immediate
data, a register, or memory (shown as S). Some instructions require the destination to be a register (shown
as R), while others allow both register and memory destinations (shown as D).

by the call tosimple _ . Here we see PC-relative addressing—the immediate 1468882 (hexadec-
imal 0x100cea ) is added to the address of the following instruction to@et00cea + 0x400516 =
0x501200 .

Figure 3 documents some of the data movement instructicaitable with x86-64, beyond those found in
IA32 (see CS:APP Figure 3.4). Some instructions requirdi¢odestination to be a register, indicated by
R. The instructions shown include different variations o thov instruction to move data into a 64-bit
register or memory destination. Moving immediate data tal#ib register can be done either with the
movq instruction, which will sign extend a 32-bit immediate v@lor with themovabsq instruction, when

a full 64-bit immediate is required.

Moving from a smaller data size to 64 bits can involve eithgn £xtension rfhovsbq, movslq ) or zero
extensionihovzbq). Perhaps unexpectedly, instructions that move or gem8eabit register values also set
the upper 32 bits of the register to zero. Consequently ibere need for an instructiomovzlq . Similarly,

the instructionmovzbq has the exact same behaviorrasvzbl when the destination is a register—both
set the upper 56 bits of the destination register to zeros Thin contrast to instructions that generate 8
or 16-bit values, such asovb; these instructions do not alter the other bits in the registhe new stack
instructionspushq andpopq allow pushing and popping of 64-bit values.

Practice Problem 1:

The following C function converts an argument of tygre _t to a return value of typdst _t , where
these two types are defined usitypedef

dest t cvt(src_t X)

{
}

Assume argument is in the appropriately named portion of registerdi (i.e.,%rdi , %edi, %di, or
%(dil ), and that some form of move instruction is to be used to pertbe type conversion and to copy

return (dest_t) x;

Instruction Effect Description

leaq S,D| D «— &S Load effective address
incq D D «— D+ 1 | Increment

decq D D — D-1 Decrement

negq D D « -D Negate

notq D D~ "D Complement

addq S,D| D «— D+ S | Add

subq S,D|D « D- S | Subtract

imulq  S,D | D « D=+ S | Multiply

xorq S,D|D « D"~ S | Exclusive-or

org S,D|D «— D| S |Or

andq S,D|D «— D& S | And

salq k,D | D «— D << k| Leftshift

shiqg k,D | D «— D << k| Leftshift (same asalq )
sarq k,D | D «— D >> k| Arithmetic right shift
shrq k,D | D «— D >> k| Logical right shift

Figure 4: Integer 64-bit arithmetic operations. They resemble their 32-bit counterparts (CS:APP Fig-
ure 3.7).

the value to the appropriately named portion of regigteax . Fill in the following table indicating the
instruction, the source register, and the destinatiorstegior the following combinations of source and
destination type:

Tx Ty Instruction S D
long long movq %rdi %rax
int long
char long

unsigned int unsigned long

unsigned char unsigned long
long int

unsigned long unsigned

As shown in Figure 4, the arithmetic and logical instrucsidar 64-bit data resemble their 32-bit counter-
parts (see CS:APP Figure 3.7). For example, we find the itt&truaddq in addition toaddl , andleaq

in addition toleal . As mentioned earlier, instructions that generate 32dgister results, such asldl ,
also set the high-order bits of the register to 0.

When mixing 32 and 64-bit data,ccmust make the right choice of arithmetic instructions, sigtensions,

and zero extensions. These depend on subtle aspects obtypersion and the interactions between the 32
and 64-bit instructions. This is illustrated by the follagiC function:

1 long int gfun(int x, int y)

2 {

3 long int t1 = (long) x +vy; [/ * 64-bit addition */
4 long int t2 = (long) (x + vy); / * 32-bit addition */
5 return t1 | t2;

6 }

Assuming integers are 32 bits and long integers are 64, thadditions in this function proceed as follows.
Recall that type conversion has higher precedence thati@ddind so line 3 calls fax to be converted to
64 bits, and by operand promotignis also converted. Valugl is then computed using 64-bit addition.
On the other hand? is computed in line 4 by performing 32-bit addition and th&teading this value to
64 bits.

The assembly code generated for this function is as follows

x86-64 version of function gfun
Argunments in registers %di (x) and %si (y)

1 gfun:

2 movslqg  %edi,%rax Convert x to |ong

3 movslq  %esi,%rdx Convert y to |ong

4 addl %esi, %edi | ower bits of t2 (32-bit addition)
5 addq %rdx, %rax t1 (64-bit addition)

6 movslqg  %edi,%rdi Sign extend to get t2

7 orq %rdi, %rax Return t1 | t2

8 ret

Local valuetl is computed by first sign extending the arguments. Mlo@slq instructions on lines 2-3
take the lower 32 bits of the arguments in registéngli and%rsi and sign extend them to 64 bits in
registers¥orax and%rdx. Theaddq instruction on line 5 then performs 64-bit addition to tfet Value
t2 is computed by performing 32-bit addition on the lower 32 loit the two operands (line 4). This value
is sign extended to 64 bits on line 6 (within a single reg)sieigett2 .

Practice Problem 2:
A C functionarithprob  with arguments, b, ¢, andd has the following body:

return a *b + c=*d,;
It compiles to the following x86-64 code:

Argunents: ain %di, bin %il, cin %dx, din %cx
1 arithprob:

2 movslq  %ecx,%rcx

3 movsbl  %sil,%esi

4 imulg %rdx, %rcx

5 imull %edi, %esi

6 leal (%rsi,%rcx), %eax

7 ret

The arguments and return value are all signed integers wfustengths. Based on this assembly code,
write a function prototype describing the return and argoitgpes forarithprob

Instruction | Effect Description
imulg S | R[%rdx]: R[%rax] < S x R[%rax] Signed full multiply
mulg S | R[%rdx]: R[%rax] « S x R[%rax] Unsigned full multiply
cltg R[%rax] < SignExtend(R[%eaX) Convert%eaxto quad word
cqto R[%rdx]: R[%rax] <« SignExtend(R[%rax]) | Convert to oct word
idivq S | R[%rdx] « R[%rdx]: R[%rax] mod S; Signed divide
R[%rax]| <« R[%rdx]: R[%rax] + S
divq S | R[%rdx] « R[%rdx]: R[%rax] mod S; Unsigned divide
R[%rax] <« R[%rdx]: R[%rax] = S

Figure 5:Special arithmetic operations. These operations support full 64-bit multiplication and division,
for both signed and unsigned numbers. The pair of registers %rdx and %rax are viewed as forming a single
128-hit oct word.

4.1 Special Arithmetic Instructions

Figure 5 show instructions used to generate the full 12@4oitiuct of two 64-bit words, as well as ones to
support 64-bit division. They are similar to their 32-biucerparts (CS:APP Figure 3.9). Several of these
instructions view the combination of registétsdx and%rax as forming a 128-bibct word For example,
theimulg andmulg instructions store the result of multiplying two 64-bit wabk—the first as given by
the source operand, and the second from registeix .

The two divide instructionglivqg anddivg start with%rdx:%rax as the 128-bit dividend and the source
operand as the 64-bit divisor. They then store the quotenégisterdorax and the remainder in register
%rdx. Preparing the dividend depends on whether unsigded () or signed idivq ) division is to be
performed. In the former case, registérdx is simply set to 0. in the latter case, the instructamito is
used to perform sign extension, copying the sign biwéx into every bit of%rdx .2

Figure 5 also shows an instructicitq  to sign extend registé¥oeaxto %rax 3. This instruction is just a
shorthand for the instructiomovslqg %eaxreg,%raxreg

Aside: What if 64-bit arithmeticisn’t good enough?

Gccerunning on x86-64 supports arithmetic using 128-bit sigamed unsigned arithmetic, via data typestl28 _t
and__uint128 _t. For example, the following C code computes the 128-bit pebaf two 64-bit integers and
stores the result in a 128-bit global variable. It also me$ihe lower 64 bits as the function result. It implements
this function using the single-operand version ofithelg instruction.

[ = Global variable for 128-bit integer */
__int128 t prod;
/= Compute 128-bit product of two 64-bit integers */
long int oct_mul(long int x, long int vy)
{
prod = (__int128_t) x A

return (long) prod;

2Gas instructioncqto s calledcqo in Intel and AMD documentation.
3Instructioncltq s calledcdge in Intel and AMD documentation.

}

Integers of this size can represent numbers in the rarigé x 10*®, nearly the range of what can be represented
as a aingle-precision floating-point numbEnd Aside.

5 Control

The control instructions and methods of implementing adritansfers in x86-64 are essentially the same
as those in IA32 (CS:APP Section 3.6). Two new instructiomgq andtestq are added to compare and
test quad words, augmenting those for byte, word, and doubtd sizes (CS:APP Section 3.6.1):

Instruction Based on| Description
cmpq  S9, 51 | S1- Sy | Compare quad words
testq S9,51 | S1 & Sy | Test quad word

5.1 Conditional Move Instructions

Starting with the PentiumPro in 1995, recent generationtAB2 processors have hambnditional move
instructions that either do nothing or copy a value to a tegidepending on the values of the condition
codes. These provide an alternate way of implementing tondi operations that can be substantially
faster than using branching code. For years, these ingingchave been largely unused.cGdid not
generate code that used them, because that would prevéatdraccompatibility. The advent of x86-64
allowed the compiler writers to forgo this compatibilityguerement, and so we now find conditional moves
in compiled x86-64 code. This section describes conditiom@ves and how they are used.

We saw in the IA32 code generated Byc that conditional statements (usitifg ) and conditional ex-
pressions (using : ) were always implemented with conditional jumps, wheredbedition code values
determine whether or not a jump is taken.

As an example, consider the following general form of agsignt and conditional expression
v = test-expr? then-expr: else-expr
For IA32, the compiler generates code having a form showiméydllowing abstract code

if (I test-expy
goto else;
v = then-expr
goto done;
else:
v = else-expr
done:

This code contains two code sequences—one evalutitergexprand one evaluatinglse-expr A combi-
nation of conditional and unconditional jumps is used taiemshat just one of the sequences is evaluated.

As noted in the discussion of branch prediction and mispegitin penalties (CS:APP Section 5.12), modern
processors can execute code of the style shown above dffjobety in the case where they can predict the
outcome of the test with high probability. The processorsuseleep pipeline that follows the predicted
branch direction, executing the chosen code sequence,cagsr&known aspeculative executionlf it
predicted the test outcome correctly, themammitsany speculative results and proceeds with no loss of
efficiency. If the prediction is incorrect, it must discatitspeculative results and restart execution with
the other code sequence. This can incur a significant detalyaps 10—40 clock cycles. Processors employ
sophisticated branch prediction hardware that attempetead any regular pattern for a given branch being
taken or not taken.

As an extreme example of this inefficiency, consider theofwihg C function for computing the maximum
of two values

int max(int x, int y)

{
}

return (X < y) ? vy : X

In a typical application, the outcome of the test< y is highly unpredictable, and so even the most
sophisticated branch prediction hardware will guess ctigrenly around 50% of the time. In addition,
the computations performed in each of the two code sequercesre only a single clock cycle. As a
consequence, the branch misprediction penalty dominagegerformance of this function. Running on an
Intel Pentium 4 Xeon, we find the function requires around yiddes per call when the outcome is easily
predictable, but around 31 cycles per call when the outcamaridom. From this we can infer that the
branch misprediction penalty is around 42 clock cycles.

Aside: How did you determinethis penalty?

Assume the probability of misprediction s the time to execute the code without mispredicatiofdsc, and the
misprediction penalty i, p. Then the average time to execute the codg,is = (1-p)Tox+p(Tox+Tmp) =
Tox + pTump. Soforp = 0.5, Tox = 10, andT,.y = 31, we getTap = 42. End Aside.

An alternate method of implementing some forms of cond#lomperations is to useonditional move
instructions. With this approach, both theen-exprand theelse-exprare evaluated, with the final value
chosen based on the evaluatiest-expr This can be described by the following abstract code:

vt = then-expr
v = else-expr
if ( testexpy v = Vvt

The final statement in this sequence illustrates a conditimove—valuest is moved tov only if the tested
condition holds.

Figure 6 illustrates some of the conditional move instautdiadded to the 1A32 instruction set with the in-
troduction of the PentiumPro microprocessor. Each of tiesteuctions has two operands: a source register
or memory locatior, and a destination registér. As with the differentset (CS:APP Section 3.6.2) and
jump instructions (CS:APP Section 3.6.3), the outcome of thiesgtLictions depend on the values of the

Instruction Synonym | Move condition Description

cmove S D | cmovz ZF Equal / zero

cmovne S D | cmovnz “ZF Not equal / not zero

cmovs SD SF Negative

cmovns S D "SF Nonnegative

cmovg S D | cmovnle |7 (SF"OF)&™ZF | Greater (signed)

cmovge S D | cmovnl " (SF"OF) Greater or equal (signeek)
cmovl S D | cmovnge | SF™OF Less (signec)

cmovle S D | cmovng (SF"OF) | ZF Less or equal (signed=)
cmova S D | cmovnbe | "CF &"ZF Above (unsigned)

cmovae SD | cmovnb | "CF Above or equal (Unsigned=)
cmovb S D | cmovnae | CF Below (unsigneck)

cmovbe SD | cmovna | CF|ZF below or equal (unsigned=)

Figure 6: The cnov instructions. These instructions copy the source value Sto its destination D when
the move condition holds. Some instructions have “synonyms,” alternate names for the same machine
instruction.

condition codes. The source value is read from either memiotlye source register, but it is copied to the
destination only if the specified condition holds.

The source and destination values can be 16, 32, or 64 bis I8ingle byte conditional moves are not
supported. Unlike the unconditional instructions, where tperand length is explicitly encoded in the
instruction name (e.gmovw, movl , movq), conditional move operand lengths can be inferred from the
names of the destination registers, and so the same instru@me can be used for all operand lengths.

As an examplegcc generates the following code for functiomax

x86- 64 code generated for function max

X in register %di, y in %si
1 max:
2 cmpl %esi, %edi Conpare Xx:y
3 cmovge  %edi, %esi if >= then y=x
4 movl %esi, %eax set y as return val ue
5 ret

This code uses a conditional move to overwrite the regigietainingy with x whenx >y (line 3).

The conditional move can be implemented with no need to préldé outcome of the test. The processor
simply evaluates the code and then either updates the aéstiregister or keeps it the same. Running the
code on the same Intel Pentium 4 Xeon, we find it requires adincycles per call regardless of the test
outcome.

Practice Prablem 3:
In the following C functionOPis some C operator declared earlier wéithefine

int arith(int x)

{
}

return x OP 4;

When compiledGcc generates the following x86-64 code:

x86-64 inplenmentation of arith

X in register %di

1 arith:

2 leal 3(%rdi), %eax

3 cmpl $-1, %edi

4 cmovle %eax, %edi

5  sarl $2, %edi

6 movl %edi, %eax return val ue in %ax
7 ret

A. What operation i©P?
B. Annotate the code to explain how it works.

Not all conditional expressions can be compiled using darhl moves. Most significantly, the abstract
code shown earlier evaluates bditlen-exprandelse-expiregardless of the test outcome. If one of those two
expressions could possibly generate an error conditionsisteaeffect, this could lead to invalid behavior.
As an illustration, consider the following C function

int cread(int *XP)

{
}

At first, this seems like a good candidate to compile usingralitmonal move to read the value designated
by pointerxp, as shown in the following assembly code:

return (xp ? *xp : 0);

Invalid inplenmentation of function cread
Xp in register %di

1 cread:

2 xorl %eax, %eax Set 0 as return val ue

3 testq %rdi, %rdi Test xp

4  cmovne (%rdi), %eax if 10, dereference xp to get return val ue
5 ret

This implementation is invalid, however, since the demafieing ofxp by thecmovne instruction (line 4)
occurs even when the test fails, causing a null pointer degating error. Instead, this code must be
compiled using branching code.

A similar case holds when either of the two branches causateaeffect, as illustrated by the following
function

int lcount = 0O;
int se_max(int x, int y)

{
}

return (x < y) ? (lcount++, y) : x;

This function increments global variabieount as part othen-expr Thus, branching code must be used
to ensure this side effect only occurs when the test comdhta@ds.

Using conditional moves also does not always improve coiilgezicy. For example, if either theen-expr
or the else-exprevaluation requires a significant computation, then thisrefs wasted when the corre-
sponding condition does not hold. Our experiments witht indicate that it only uses conditional moves
when the two expressions can each be computed with a sirggteétion. This is being too conservative for
current processors, given their high branch mispredigiemalties.

5.2 Looping Constructs

In CS:APP Section 3.6.5, we found that the three looping tcoats of C:do-while , while , andfor
were all compiled for IA32 using a common template based erstructure oflo-while . The other loop
forms were transformed intdo-while as illustrated in CS:APP Figure 3.14.

With x86-64, we find a richer variety of loop templates. Thepéate fordo-while remains the same,
and somevhile andfor statements are implemented in this form. For example, tde generated for
fib _wandfib _f shown in CS:APP Section 3.6.5 follows the same pattern dsstimvn for IA32. In
some cases, howevevhile andfor loops are translated using a different template. Let us exathese
two different loop templates.

As an illustration, consider the following C function forraputing factorial using do-while loop

int fact_dw(int x)

{
int result = 1,
do {
result *= x;
X==
} while (x > 0);
return result;
}

Gcc generates the following code for x86-64

x86-64 inplenmentation of fact_dw
X in register %di

1 fact_dw:

2 movl $1, %eax result =1

3 .L2: | oop:

4 imull %edi, %eax result *= x
5 decl %edi X- -

testl %edi, %edi Test x
jg9 L2 if >0 goto | oop
rep ; ret el se return

This code follows the loop template seen earlier. Withinlttog, theimull  anddecl instructions imple-
ment the body, while theestl andjg instructions implement the test. The loop is always enté&@a
the top. The general form of this implementation is

loop:

body-statement
t = test-expr

i (t)

goto loop;

Aside: Why istherear ep instruction in this code?

On line 8 of the x86-64 code fdiact _dw, we see that the procedure ends with the instruction cortibma
rep; ret ,ratherthan simplyet . Looking at the Intel and AMD documentation for trep instruction, we see
that it is normally used to implement a repeating string apen [3, 6]. It seems completely inappropriate here.
The answer to this puzzle can be seen in AMD’s guidelines mapiler writers [1]. They recommend this particular
combination to avoid making theet instruction be the target of a conditional jump instructidrnis is the case
here, because it is preceded by g ainstruction, and in the event the jump condition does notihtfle program
“falls through” to the return. According to AMD, the processloes a better job predicting the outcome of the
branch if it does not have @t instruction as a target. Thigp instruction will have no effect, since it is not
followed by a string manipulation instruction. Its only pose is to serve as a branch targatd Aside.

Now consider a factorial function based owhile loop:

int fact_while(int x)

{

}

int result = 1;

while (x > 0) {
result *= Xx;
X--;

}

return result;

(These two functions yield different results wher< 0.) Gcc generates the following code for x86-64

o g~ W N PP

x86-64 inplenentation of fact_while
X in register %di

fact_while:
movl $1, %eax result =1
jmp .L12 goto niddle
.L13: | oop:
imull %edi, %eax result *= x
decl %edi X- -

7 .L12: m ddl e:

8 testl %edi, %edi Test x
9 g .L13 if >0 goto | oop
10 rep ; ret el se return

In this loop, we see the exact same four instructions, butla® see a second entry point into the loop,
labeledmiddle inthe comments. When entering at this point, only the irsimns implementing the loop
test are executed. The program enters via this point at #neaftthe loop, providing the initial loop test.
This code has the following general form:

goto middle
loop:

body-statement
middle:

t = test-expr

if (t)

goto loop;

Comparing this form to the form based on translativigile into do-while , we see that this new code
requires an additional unconditional jump at the beginning it then avoids having duplicate copies of the
test code. On a modern machine, unconditional jumps havitdg performance overhead, and so this
new form would seem better in most cases.

Practice Problem 4:
For the following C function, the expressioBXPRITHEXPR5were specified usingdefine

long int puzzle(int a, int b)

{ . .
int i
long int result = EXPR1;
for (i = EXPR2; i > EXPR3; i -= EXPR4)
result *= EXPRS5;
return result;
}

Gcc generates the following x86-64 code:

x86-64 inplenmentation of puzzle
ain register %di, b in register %esi
return value in register % ax

1 puzzle:

2 movslq  %esi,%rdx
3 jmp .L60

4 .L61:

5 movslq  %edi,%rax
6  subl %esi, Yedi
7 imulqg %rax, %rdx

8 .L60:
9 testl %edi, %edi

10 g .L61
11 mov(q %rdx, %rax
12 ret

A. What register is being used for local variabésult  ?
B. What register is being used for local variabfe
C. Give valid definitions for expressioBEXPRH-EXPR5

6 Procedures

We have already seen in our code samples that the x86-64rimaptation of procedure calls differs sub-
stantially from that of IA32. By doubling the register setpgrams need not be so dependent on the stack
for storing and retrieving procedure information. This ¢meatly reduce the overhead for procedure calls
and returns.

Here are some of the highlights of how procedures are impiéedewith x86-64:

6.1

Arguments (up to the first six) are passed to procedures giategs, rather than on the stack. This
eliminates the overhead of storing and retrieving valuetherstack.

Thecallg instruction stores a 64-bit return pointer on the stack.

Many functions do not require a stack frame. Only functidmest tannot keep all local variables in
registers need to allocate space on the stack.

Functions can access storage on the stack up to 128 byteséym, at a lower address than) the
current value of the stack pointer. This allows some fum&im store information on the stack without
incrementing or decrementing the stack pointer.

There is no frame pointer. Instead, references to stackitmsaare made relative to the stack pointer.
Typical functions allocate their total stack storage nedlse beginning of the call and keep the stack
pointer at a fixed position.

As with IA32, some registers are designated as callee-sgisters. These must be saved and restored
by any procedure that modifies them.

Argument Passing

Up to six integral (i.e., integer and pointer) arguments loaipassed via registers. The registers are used in
a specified order, and the name used for a register depentie sizé of the data type being passed. These
are shown in Figure 7. Arguments are allocated to theseteegiaccording to their ordering in the argument

Argument Operand Size (bits)

Number 64 32 16 8
1 %rdi %edi %di %dI
2 %orsi %esi %si %sil
3 %rdx %edx %dx %dI
4 %rcx %ecx %cCX %cl
5 %r8 %r8d %r8w %r8b
6 %r9 %r9d %row %r9b

Figure 7: Registers for passing function arguments The registers are used in a specified order and
named according to the argument sizes.

list. Arguments smaller than 64 bits can be accessed usingppropriate subsection of the 64-bit register.
For example, if the first argument is 32 bits, it can be acckas@oedi.

As an example of argument passing, consider the followingrtfon having eight arguments

void proc(long al, long +xalp,
int a2, int *az2p,
short a3, short *a3p,
char a4, char * a4p)
{
xalp += al,
xaz2p += az;
*a3p += a3;
*adp += a4,
}

The arguments include a range of different sized integets38, 16, and 8 bits) as well as different types
of pointers, each of which is 64 bits.

This function is implemented in x86-64 as follows:

x86-64 inplenmentation of function proc
Argunment s passed as foll ows:

al in %di (64 bits)
alp in %si (64 bits)
a2 in %dx (32 bits)
azp in %cx (64 bits)
a3 in % 8w (16 bits)
a3pin %9 (64 bits)

a4 at 8(%sp) ( 8 bits)
adp at 16(% sp) (64 bits)

1 proc:
2 movq 16(%rsp), %rl0 Fetch adp (64 bits)
3 addq %rdi, (%rsi) xalp += al (64 bits)
4 addl %edx, (%rcx) xa2p += a2 (32 bits)
5 movzbl 8(%rsp), Yeax Fetch a4 (8 hits)

6 addw %r8w, (%r9) *a3p += a3 (16 bits)
7 addb %al, (%rl0) xadp += a4 (8 bits)
8 ret

The first six arguments are passed in registers, while thevimsare at positions 8 and 16 relative to the
stack pointer. Different versions of tteld instruction are used according to the sizes of the operands:
addq for al (long ), addl fora2 (int ), addw for a3 (short ), andaddb for a4 (char ).

Practice Problem 5;
A C functionincrprob  with argumentsgy, t , andx (not given in that order) has the following body:

*t += X;
*Qq += *1;

It compiles to the following x86-64 code:

1 incrprob:

2 addl (%rdx), %edi
3 movl %edi, (%rdx)
4 movslq  %edi,%rdi

5 addq %rdi, (%rsi)
6 ret

Determine all valid function prototypes forcrprob by determining the ordering and possible types
of the three parameters.

6.2 Stack Frames

We have already seen that many compiled functions do notreegstack frame. If all of the local variables
can be held in registers, and the function does not call dmgrdtinctions (sometimes referred to deaf
procedure in reference to the tree structure of procedure callsy the only need for the stack is to save
the return pointer.

On the other hand, there are several reasons a function mayae stack frame:

e There are too many local variables to hold in registers.

Some local variables are arrays or structures.

The function uses the address-of operag)rt¢ compute the address of a local variable.

The function must pass some arguments on the stack to arfottotion.

The function needs to save the state of a callee-save regeftae modifying it.

When any of these conditions hold, we find the compiled codth&function creating a stack frame. Unlike
the code for IA32, where the stack pointer fluctuates backfarid as values are pushed and popped, the
stack frames for x86-64 procedures usually have a fixed seeat the beginning of the procedure by
decrementing the stack pointer (registérsp). The stack pointer remains at a fixed position during the
call, making it possible to access data using offsets veldt the stack pointer. As a consequence, the
frame pointer (registe¥oebp) seen in I1A32 code is no longer needed.

In addition, whenever one function (tkaller) calls another (thealleg, the return pointer gets pushed onto
the stack. By convention, we consider this part of the calliack frame, in that it encodes part of the
caller’s state. But this information gets popped from tlaelstas control returns to the caller, and so it does
not affect the offsets used by the caller for accessing salithin the stack frame.

The following function illustrates many aspects of the x®bstack discipline.

long int call_proc()

{
long x1 = 1; int x2 = 2;
short x3 = 3; char x4 = 4;
proc(xl, &x1, x2, &x2, x3, &x3, x4, &x4);
return (x1+x2) * (x3-x4);
}

Gcc generates the following x86-64 code.

x86- 64 inplenmentation of call_proc

1 call_proc:

2 subq $32, %rsp Allocate 32-byte stack frame
3 mov! $2, %edx Pass 2 as argunment 3
4 mov! $3, %r8d Pass 3 as argument 5
5 leaq 31(%rsp), Yorax Conpute &x4

6 leaq 24(%rsp), %rcx Pass &2 as argument 4
7 leaq 28(%rsp), %r9 Pass &3 as argument 6
8 leaq 16(%rsp), %rsi Pass &1 as argument 2
9 mov! $1, %edi Pass 1 as argunent 1
10 movq $1, 16(%rsp) x1 =1

11 movq %rax, 8(%rsp) Pass &4 as argument 8
12 movl $2, 24(%rsp) X2 = 2

13 movw $3, 28(%rsp) x3 = 3

14 movb $4, 31(%rsp) X4 = 4

15 movl $4, (%rsp) Pass 4 as argument 7
16 call proc Cal |

17 movswl  28(%rsp),%edx Get x3

18 movsbl  31(%rsp),%ecx Get x4

19 movslq  24(%rsp),%rax Get x2

20 addq 16(%rsp), %rax Conput e x1+x2

21 addq $32, %rsp Deal | ocate stack frane
22 subl %ecx, %edx Conpute (int) (x3-x4)
23 movslq  %edx,%rdx Sign extend to |ong

24 imulq %rdx, %rax Return (x1+x2)=*(x3-x4)
25  ret

A). Before call toproc

31 28
x4 x3 X2
24
x1
16
Argument 8 8
Stack pointer Argument 7 0
rsp
B). During call toproc
x4 x3 X2
x1
Argument 8
g 16
Argument 7
g 8
Stack pointer Return Pointer 0
¥rsp

Figure 8:Stack frame structure for cal | _pr oc. The frame is required to hold local variables x1-x4, as
well as for the seventh and eighth arguments to proc . During the execution of proc (B), the stack pointer
is shifted down by 8.

Figure 8A illustrates the stack frame set up during the exacwf call _proc . Functioncall _proc
allocates 32 bytes on the stack by decrementing the stankepoit uses bytes 16—-31 to hold local variables
x1 (bytes 16-23)x2 (bytes 24-27)x3 (bytes 28-29), anat4 (byte 31). These allocations are sized
according to the variable types. Byte 30 is unused. Bytesahe/8—15 of the stack frame are used to
hold the seventh and eighth argumentsati _proc , since there are not enough argument registers. The
parameters are allocated eight bytes each, even thougm@@rx4 requires only a single byte. In the
code forcall _proc , we can see instructions initializing the local variabled aetting up the parameters
(both in registers and on the stack) for the calc&dl _proc . After proc returns, the local variables are
combined to compute the final expression, which is returneegiste®rax . The stack space is deallocated
by simply incrementing the stack pointer before tbe instruction.

Figure 8B illustrates the stack during the executiorpofc . Thecall instruction pushed the return
pointer onto the stack, and hence the stack pointer is dhiftevn by 8 relative to its position during the
execution ofcall _proc . Hence, within the code fgiroc , arguments 7 and 8 are accessed by offsets of
8 and 16 from the stack pointer.

Observe howcall _proc changed the stack pointer only once during its executionc Getermined that
32 bytes would suffice for holding all local variables and falding the additional arguments proc .
Minimizing the amount of movement by the stack pointer sifigd the compiler’s task of generating refer-
ence to stack elements using offsets from the stack pointer.

6.3 Register Saving Conventions

We saw in IA32 (CS:APP Section 3.73) that some registers fmsdtblding temporary values are by desig-
nated asaller savedwhere the callee is free to overwrite their values, whileeos arecallee savegwhere
the callee must save their values on the stack before wiitinlgem. With x86-64, the following registers
are designated as being callee saédox, %rbp, and%r12-%r15.

Aside: Arethere any caller-saved temporary registers?

Of the 16 general-purpose registers, we've seen that sidesignated for passing arguments, six are for callee-
saved temporaries, on#(ax) holds the return value for a function, and of&rép) serves as the stack pointer. In
addition,%r10 has a specific use in supporting languages that alatic scoping This includes Pascal, but not C
or C++. Only%r12 is left as a caller-saved temporary register, but this isgaly reserved for use by the linking
code.

Of course, some of the argument registers can be used whecedpre has less than six arguments, Znax can
be used within a procedure. Moreover, it's always possibkat/e the state of some callee-saved registers and then
use them to hold temporary valudsnd Aside.

We illustrate the use of callee-saved registers with a sdmewnusual version of a recursive factorial
function:

/* Compute x! and store at resultp * [
void sfact_helper(long int x, long int * resultp)
{
if (x <= 0)
xresultp = 1;
else {

A). Before decrementing the stack pointer (on line 4)
Stack pointer

3rsp P
Saved $rbp
-8
Saved %rb
aved $rbx _16
B). After decrementing the stack pointer
Saved $rb
\"{ rop 16
Saved $rb
\"{ rox 8
Stack pointer nresult 0
$rsp

Figure 9: Stack frame for function sf act _hel per. This function decrements the stack pointer after
saving some of the state.

long int nresult;

sfact_helper(x-1, &nresult);

xresultp = x * nresult;
}

To compute the factorial of a value this function would be called at the top level as follows:

long int sfact(long int X)

{
long int result;
sfact_helper(x, &result);
return result;

}

The x86-64 code fosfact _helper is shown below

x86-64 inplenmentation of sfact_hel per
Argument x in %di, resultp in %si
1 sfact_helper:

2 movq %rbx, -16(%rsp) Save % bx (callee save)

3 movq %rbp, -8(%rsp) Save % bp (callee save)

4  subqg $24, %rsp Allocate 24 bytes on stack
5 testq  %rdi, %rdi Test x

6 movq %rdi, %rbx Copy x to 9% bx

7 movq %rsi, %rbp Copy resultp to % bp

8 jle .L17 if x<=0 goto finish

9 leaq -1(%rdi), %rdi xml = x-1 (1st argument)
10 movq %rsp, %orsi &iresult  (2nd argunent)
11 call sfact_helper sfact _hel per (xmil, &nresult)
12 imulq (%rsp), %rbx x*nresul t

13 movq %rbx, (%rbp) xresult = x+nresult

14 movq 8(%rsp), %rbx Rest ore 9% bx

15 movq 16(%rsp), %rbp Restore % bp

16 addq $24, %rsp Deal | ocate stack frame
17 ret Return

18 .L17: finish:

19 movq $1, (%orsi) xresultp = 1

20 movq 8(%rsp), %rbx Rest ore 9% bx

21 movq 16(%rsp), %rbp Restore 9% bp

22 addq $24, %rsp Deal | ocate stack frame
23 ret Return

Figure 9 illustrates howfact _helper uses the stack to store the values of callee-saved registdro
hold the local variabl@result . This implementation has the interesting feature thatwledallee-saved
registers it useprbx and%rbp) are saved on the stack (lines 2-6&¥orethe stack pointer is decremented
(line 4) to allocate the stack frame. As a consequence, Huk siffset for%rbx shifts from —16 at the
beginning to+8 at the end (line 14). Similarly, the offset fétrbp shifts from—8 to +16.

Being able to access memory beyond the stack pointer is asuahteature of x86-64. It requires that the
virtual memory management system allocate memory for #gibn. The x86-64 ABI [8] specifies that
programs can use the 128 bytes beyond (i.e., at lower addréisan) the current stack pointer. The ABI
refers to this area as thred zone It must be kept available for reading and writing as thelstaainter
moves.

Note also how the code fafact _helper has two separateet instructions: one for each branch of
the conditional statement. This contrasts with 1A32 codbeng we always saw the different branches
come together at the end of a function to a shaetd instruction. This is partly because 1A32 requires a
much more elaborate instruction sequence to exit a funciod it is worthwhile to avoid duplicating these
instructions at multiple return points.

Practice Problem 6:
For the following C program

long int local_array(int i)

{
long int a[4] = {2L, 3L, 5L, 7I};
int idx =i & 3;
return a[idx];

}

Gccgenerates the following code

x86-64 inplementation of |ocal_array
Argurment i in %edi

1 local_array:

2 andl $3, %edi

3 movq $2, -40(%rsp)

4 movq $3, -32(%rsp)

27



5  movq $5, -24(%rsp)

6  movq $7, -16(%rsp)

7 mov(q -40(%rsp,%rdi,8), %rax
8 ret

A. Draw a diagram indicating the stack locations used byftimstion and their offsets relative to the
stack pointer.

B. Annotate the assembly code to describe the effect of eetiuction
C. What interesting feature does this example illustratiaithe x86-64 stack discipline?

Practice Problem 7:
For the following recursive factorial program

long int rfact(long int x)

{
if (x <= 0)
return 1;
else {
long int xml = x-1;
return x  * rfact(xml);
}
}

Gccgenerates the following code

x86-64 inplenmentation of recursive factorial function rfact
Argunment x in %di
1 rfact:

2 testq  %rdi, %rdi

3 pushq %rbx

4 movl $1, %eax
5 mov(q %rdi, %rbx
6 jle L9

7 leaq -1(%rdi), %rdi
8 call rfact

9 imulqg %rbx, %rax
10 .L9:

11 popq %rhbx

12 ret

A. What value does the function store%trbx ?

B. What are the purposes of theshqg (line 3) andpopq (line 11) instructions?

C. Annotate the assembly code to describe the effect of @attuction

D. What interesting feature does this example illustratauathe x86-64 stack discipline?

7 Data Structures

Data structures follow the same principles in x86-64 as tlein IA32: arrays are allocated as sequences of
identically-sized blocks holding the array elements,atriies are allocated as sequences of variably-sized
blocks holding the structure elements, and unions areaildcas a single block big enough to hold the
largest union element.

One difference is that x86-64 follows a more stringent setligihment requirements. For any scalar data
type requringK bytes, its starting address must be a multipléofThus, data typekng , double , and
pointers must be aligned on 8-byte boundaries. In additatg typdong double uses a 16-byte align-
ment (and size allocation), even though the actual reptaisem requires only 10 bytes. These alignment
conditions are imposed to improve memory system perforerartbe memory interface is designed in most
processors to read or write aligned blocks that are eighikt@en bytes long.

Practice Problem 8:

For each of the following structure declarations, detesire offset of each field, the total size of the
structure, and its alignment requirement under x86-64.

A. struct P1 { int i; char c; long j; char d; };

B. struct P2 { long j; char c; char d; int i; };

C. struct P3 { short w[3]; char c[3] };

D. struct P4 { short w[3]; char *C[3] };

E. struct P3 { struct P1 a[2]; struct P2 *p }

8 Floating Point

Starting with the Pentium MMX in 1997, both Intel and AMD hawdroduced successive generations
of mediainstructions to support graphics and image processing. VErgon of these supported l®cc
for x86-64 is is name®&SE3 for “Streaming SIMD Extensions, version 3.” It is availalin all x86-64
processors. The media instructions originally focused|mwang multiple operations to be performed in
a parallel mode known asngle instruction, multiple datar SIMD (pronounced SIM-DEE). In this mode
the same operation is performed on a number of different\gdiees in parallel.

The media instructions implement SIMD operations by hawrsgt of registers that can hold multiple data
values inpackedformat. For example, SSE3 provides sixte¢hlM registers of 128 bits each, named
%xmme%xmm15 Each one of these registers can hold a vectoK aflements ofNV bits each, such that

K x N = 128. For integers/{ can be 8, 16, 32, or 64 bits, while for floating-point numbéfs;an be 32

or 64. For example, a single SSE3 instruction can add two\mdtors of eight elements each. The floating
point formats match the IEEE standard formats for single @mable-precision values. The major use of
these media instructions are in library routines for greplaind image processing. These routines must be
written in assembly code, or by using special extensionssapgported byscc.

With the SSE3 came the opportunity to completely change tiefleating point code is compiled for x86
processors. Historically, floating point is implementedhvan arcane set of instructions dating back to the

Instruction Source | Destination| Description

Mmovss Msol X X Move single precision

Movss X Mso Move single precision

movsd Mgyl X X Move double precision

movsd X Mgy Move double precision

cvtss2sd Msol X X Convert single to double precision

cvtsd2ss Mgyl X X Convert double to single precision

cvtsi2ss M3o/R3o X Convert integer to single precision

cvtsi2sd Msa/R3o X Convert integer to double precision

cvtsi2ssq Mgyl Rgy X Convert quadword integer to single precision

cvtsi2sdq Mgyl Rgy X Convert quadword integer to double precision

cvttss2si XIMso Rso Convert with truncation single precision to integer
cvitsd2si X/ Mg, R3o Convert with truncation double precision to integer
cvitss2siq X/ Mso Rgy Convert with truncation single precision to quadword irteg
cvitsd2siq X/ Mg, Rgq Convert with truncation double precision to quadword ieteg

X: XMM register (e.g.%xmmp

R3o: 32-bit general-purpose register (e Yheax)
Rg4: 64-bit general-purpose register (e &rax)
Mss: 32-bit memory range

Mgy 64-bit memory range

Figure 10: Floating-point movement and conversion operations. These operations transfer values
between memory and registers, possibly converting between data types.

8087, a floating-point coprocessor for the Intel 8086. Thiesiuctions, often referred to as “x87” operate
on a shallow stack of registers, a difficult target for opimg compilers (CS:APP Section 3.14). They also
have many quirks due to a nonstandard 80-bit floating-poimhét (CS:APP Section 2.4.6).

SSES3 introduced a set etalarfloating-point operations that work on single values in the-brder 32 or
64 bits of XMM registers, The overall effect is to provide & skregisters and instructions that are more
typical of the way other processors support floating poirtc@ow implements floating point using mostly
SSE3 instructions, reverting to x87 floating point for onliea special cases.

This section describes the x86-64 implementation of flggtioint based on SSE3. It supplants the presen-
tation of floating point from CS:APP, Section 2.4.5.

8.1 Floating-Point Movement and Conversion Operations

Figure 10 shows a set of instructions for transfering dathfanperforming conversions between floating-
point and integer data types. Floating-point data are hiiheérein memory (indicated in the table ags,
andMg,) or in XMM registers (shown in the table &S). Integer data are held either in memory (indicated
in the table as\/3, or Mg,4) or in general-purpose registers (shown in the tabl&-asand Rg4). All 32-bit
memory data must satisfy a 4-byte alignment, while 64-hi&a aiaust satisfy an 8-byte alignment.

The floating-point move operations can transfer data frayister to register, from memory to register and
from register to memory. A single instruction cannot moveadeom memory to memory. The floating-
point conversion operations have either memory or a reggstsource and a register as destination, where
the registers are general-purpose registers for integarasel XMM registers for floating-point data. The
instructions, such asvttss2si , for converting floating-point values to integers use tation, always
rounding values toward zero, as is required by C and most ptegramming languages.

As an example of the different floating-point move and cosiggr operations, consider the following C
function:

double fevt(int i, float *fp, double +xdp, long =*Ip)
{
float f = +fp; double d = xdp; long | = *p;
*Ip = (long) d;
xfp = (float) i;
+dp = (double) I;
return (double) f;
}

and its associated x86-64 assembly code

x86-64 inpl mentation of fcvt

Argunent s:
i %edi int
fp % si float *
dp % dx doubl e *
Ip % cx | ong *
1 fevt:
2 movipd (%rdx), %xmmO d = «dp
3 movq (%rcx), %r8 I =+lp
4 movss  (%rsi), %xmml f = xfp
5 cvttsd2siq %xmmO0, %rax (1ong) d
6 Xorps %xmm0, %xmmO Set %m0 to O
7 cvtsi2ss %edi, %xmm0 (float) i
8 movq %rax, (%rcx) *Ip = (long) d
9 movss %xmmO, (%orsi) *fp = (float) i

10
11
12
13

All of the arguments tdcvt

%r8, %xmmO0
%xmmO, (%rdx)
%xmm1l, %xmmoO

cvtsi2sdq
movsd
cvtss2sd
ret

(double) |
*dp = (double) I
Return (double) f

are passed through the general-purpose registers, sacartheither integers

or pointers. The return value is returned in regi$tetrmmpthe designated return register filvat or
double values. In this code, we see a number of the move and conmersitvuctions of Figure 10. In line
2, we see a variant of thmovsd instruction calledmovlpd . This “move lower packed double-precision
floating point” differs frommovsd in its affect on the upper 64 bits of the destination XMM régisSince
we are not making use of these bits, the two instructions eamskd interchangeably.

In line 6 we see the instructioxorps .
register%xmma@o all zeros, much as we've seen tkerl

In this context, the instruction is simply setting the desion

used to set an integer register to zero (see

Single | Double | Effect Description

addss addsd | D «— D + S Floating-point add

subss subsd | D « D — S Floating-point subtract
mulss mulsd | D « D x S Floating-point multiply
divss divsd D« D/S Floating-point divide
maxss | maxsd | D < max(D, S) | Floating-point maximum
minss minsd | D « min(D, S) | Floating-point minimum
sqrtss sqrtsd D — /§ Floating-point square rogt

Figure 11:Floating-point arithmetic operations. All have source S and destination D operands.

CS:APP Practice Problem 3.6). It's not clear whgc inserted this instruction, since the lower 32 bits of
the register are set to the value(@bat) i by the next instruction, and only those bits are required by
the subsequent instructionmigvss andmovsd) having%xmm@s source.

Practice Problem 9:

The following C function converts an argument of tygre _t to a return value of typdst _t , where
these two types are defined usitypedef

dest t cvt(src_t Xx)

{
}

Assume argument is either in%xmm@r in the appropriately named portion of registérdi (i.e.,
%rdi or %edi), and that one of the conversion instructions is to be uspétfmrm the type conversion
and to copy the value to the appropriately named portion gister %rax (integer result) ooxmmO
(floating-point result). Fill in the following table inditiag the instruction, the source register, and the
destination register for the following combinations of smuand destination type:

return (dest_t) x;

Tx Ty Instruction S D
long double | cvtsi2sdq %rdi %xmmO
double int
float double
long float
float long

8.2 Floating-Point Arithmetic operations

Figure 11 documents a set of SSE3 floating-point instrusttbat perform arithmetic operations. Each has
two operands: a sourcg, which can be either an XMM register or a memory location, anéstinationD,
which must be an XMM register. Each operation has an instnudor single-precision and an instruction
for double precision. The result is stored in the destimatagister.

As an example, consider the example function from CS:APR@e8.14.5:

double funct(double a, float

{
}

return a *Xx - bl

X, double b, int i)

The x86-64 code is as follows:

x86-64 inpl enmentation of funct
Argunent s:
a % D doubl e
X % mmil f1 oat
b % doubl e
i Y%edi int
1 funct:
2 cvtss2sd %xmm1, %xmml onmil = (double) x
3 mulsd %xmmO0, %xmml ol = a*x
4 cvtsi2sd %edi, %xmm0 %m0 = (double) i
5 divsd %xmm0, %xmm2 oxmm2 = b/i
6 movsd %xmm1, %xmmO0 oxmmD = axx
7 subsd %xmm2, %xmmO return a*x - b/i
8 ret

The three floating point argumerds x, andb are passed in XMM registefdxmme%xmm2while integer
argument is passed in registébedi. Conversion instructions are required to convert argusrerndi

to double. The function value is returned in regigfexmmOCompared to the stack code shown for IA32

in CS:APP, floating-point x86-64 code more closely resembiteger code.

Unlike integer arithmetic operations, the floating-poipemations cannot have immediate values as operands.

allocate and initialize storfogeany constant values, and then generate code
memory. This is illustrated byfttlewing Celsius to Fahrenheit conversion

Instead, the compiler must
that reads the values from
function:

double cel2fahr(double temp)

{
}

return 1.8

* temp + 32.0;

The relevant parts of the x86-64 assembly code are as fallows

Const ant decl arati ons

1 .LC2:

2 .long 3435973837

3 .long 1073532108

4 LC4:

5 .long 0

6 dong 1077936128
Code

Low order four bytes of 1.8
Hi gh order four bytes of 1.8

Low order four bytes of 32.0
Hi gh order four bytes of 32.0

x86- 64 inpl ementation of cel 2f ahr
Argument tenp in %nmD

7 cel2fahr:

8 mulsd .LC2(%rip), YoxmmO Miltiply by 1.8
9 addsd .LC4(%rip), %xmmO Add 32.0

10 ret

We see that the function reads the value 1.8 from the memoayitn labeled , and the value 32.0 from the
memory location labeled . Looking at the values associatiu these labels, we see that each is specified
by a pair of.long declarations with the values given in decimal. How shouleséhbe interpreted as
floating-point values? Looking at the declaration labelegk,see that the two values aig0x0) and0
(Ox0). Since the machine uses little-endian byte ordering, tisevialue gives the low-order 4 bytes, while
the second gives the high-order 4 bytes. From the high-dogtess, we can extract an exponent fieldg0

(0), from which we subtract a bias of 1023 to get an exponent@23. Concatenating the fraction bits of
the two values, we get a fraction field 8k0000000000000 , which can be shown to be the fractional
binary representation of 0.8, to which we add the impliedlileg one to get 1.8.

Practice Problem 10:

Show how the numbers declared at label encode the number 32.0

8.3 Floating-Point Comparison oper ations

The SSE3 extenstions provide two instructions for comggiiimating values:

Instruction Based on| Description
ucomiss  S», 51 | S1- Sy | Compare single precision
ucomisd S, 51 | S1- Sy | Compare double precision

These instructions are similar to tbenpl andcmpq instructions (see CS:APP Section 3.6.1), in that they
compare operandS; andSs and set the condition codes to indicate their relative &alues withcmpq,
they follow theGAs convention of listing the operands in reverse order. Argunsg must be in an XMM
register, whileS; can either be in an XMM register or in memaory.

The floating-point comparison instructions set three diorlicodes: the zero flag ZF, the carry flag CF, and
the parity flag PF. We did not document the parity flag in CS: Afféause it is not used &cc-generated
IA32 code. For integer operations, this flag is set when thstmezent arithmetic operation yielded a value
with odd parity (i.e., an odd number of 1's in the word). Forfing-point comparisons, however, the flag
is set when either operand aN By convention, any comparison is considered to fail whea ohthe
arguments is &aN, and this flag is used to detect such a condition. For exanepkn the comparison
X == X vyields 0 wherx is aNaN

The condition codes are set as follows:

Ordering | CF ZF PF
Unordered| 1 1 1
<

1 0 0
= 0 1 0
> 0 0 0

The Unorderedcase occurs when either of the operandda$\l. This can be detected from the parity flag.
Commonly, thgp (for “jump on parity”) instruction is used to conditionaljymp when a floating-point
comparison yields an unordered result. Except for this,dasevalues of the carry and zero flags are the
same as those for an unsigned comparison: ZF is set when ¢thepevands are equal, and CF is set when
S1 < Ss. Instructions such ga andjb are used to conditionally jump on various combinations ebth
flags.

As an example of floating-point comparisons, the followinfu@ction classifies its argumertaccording
to its relation to 0.0, returning an enumerated type astesul

typedef enum {NEG, ZERO, POS, OTHER} range_t;

range_t find_range(float x)

{
int result;
if (x < 0)
result = NEG;
else if (x == 0)
result = ZERO;
else if (x > 0)
result = POS;
else
result = OTHER;
return result;
}

Enumerated types in C are encoded as integers, and so tliel@dssction values are: NEG, 1 (ZERQ,
2 (POSY, and 3 OTHER This final outcome occurs when the valuexas NaN.

Gcc generates the following x86-64 code ford _range :

x86- 64 inpl enmentation of find_range
Argunment x in %m0 (single precision)

1 find_range:

2 Xorps %xmm1, %xmml ol = 0

3 xorl %eax, %eax result =0

4 ucomiss %xmmO0, %xmm1l Conpare 0: x

5 ja .L5 If (> and not NaN) goto done
6 movb $1, %al result = 1

7 ucomiss %xmm1l, %xmmO Conpare x:0

8 jp .L13 If unordered goto next

9 je .L5 If (= and not NaN) goto done
10 .L13: next :

11 xorl %eax, %eax result = 0

12 ucomiss %xmm1, %xmmO conpare x: 0

13 setbe %al if (<= or NaN) result =1
14 addl $2, %eax result += 2

15 .L5: return

16 rep ; ret

The code is somewhat arcane—it comparés 0.0 three times, even though the required informatiordcou
be obtained with a single comparison. Let us trace the flowmefunction for the four possible comparison
results.

X < 0.0: Theja instruction on line 5 will be taken, jumping to the end withedurn value of 0.

x = 0.0: Theja instruction on line 5 will not be taken, nor will be thg instruction on line 8. Th¢e
instruction on line 9 will be taken, jumping to the end withedurn value of 1 (set on line 6).

x > 0.0: Neither theja instruction on line 5, thgp instruction on line 8, nor thgg instruction on line 9
will be taken. In addition, theetbe instruction on line 13 will not change the return value, aad s
the return value will be incremented from 0 to 2 on line 14, #ns value will be returned.

X = NaN: Theja instruction on line 5 will not be taken, but the instruction on line 8 will be. Then
thesetbe instruction on line line 13 will change the return value frOno 1, and this value is then
incremented from 1 to 3 on line 14.

Once again, we see that x86-64 floating-point code more lglessembles integer code than it does the
stack-based floating-point code of IA32.

8.4 Floating-Point Code in Procedures

We have already seen examples of several functions thaffloatimg-point arguments and a floating-point
result. The XMM registers can be used for these purposescifi®jadly, the following conventions are
observed:

e Up to eight floating point arguments can be passed in XMM tegg&xmme%xmm./These registers
are used in the order the arguments are listed. Additionatifig-point arguments can be passed on
the stack.

e A function that returns a floating-point value does so inseg%xmm0o

e All XMM registers are caller saved, The callee may overwdte/ of these registers without first
saving it.

8.5 Performing Common Floating-Point Operationsin Uncommon Ways
At times, Gcc makes surprising choices of instructions for performingiowmn tasks. As examples, we've

seen how théeal instruction is often used to perform integer arithmetic (8&F Section 3.5), and the
xorl instruction is used to set registers to 0 (CS:APP Practiobl®m 3.6).

There are far more instructions for performing floatingrp@perations than we've documented here or in
CS:APP Section 3.14, and some of these appear in unexpdatas pWe document a few such cases here.

The following assembly code shows examples of instructoe®gned to operate on vectors of values in
packed format:

movlpd (%rax), %xmmO Read doubl e into 9%nmD
X0rps %xmm2, %xmm2 Set %R to zero
movaps %xmmO0, %xmml Copy %m0 to Y%mmil

As mentioned earlier, thenovlpd is similar to themovsd instruction. In this example, it reads the eight
bytes starting at the address specified by reg¥terx and stores it in the lower eight bytes of XMM register
%xmm0Q

Thexorps instruction is an exclusive-or instruction for XMM data.n§é both operands are the same in
this example, the effect is to s&xmm2o 0.

Themovaps instruction (for “move aligned packed single-precisiois’used to copy entire XMM registers.
In this example, it simply copie®xmm@o %xmml
In addition, there are times thatcuses IA32 floating-point code in x86-64. Most commonly, ikesuse

of the ability of IA32 floating-point instructions to perforformat conversions while loading a value from
memory and while storing a value to memory. This is illugidaby the following code sequence

flds 16(%rsp) Load single-precision into FP register
fstpl  8(%rsp) Store FP register as double precision

Recall that 1A32 floating point makes use of a shallow stackegfsters, each holding an 80-bit floating-
point value. In this example, tHids instruction reads a single-precision value from memorgyveds it
to the 80-bit format, and stores the result in the top stagister. Thefstpl  instruction pops the top value
from the stack, converts it to double precision, and stdariesrmemory.

This combination of copying and conversion could readilyrbplemented using SSE instructions:

MmMovss 16(%rsp), %xmmo0 Read singl e-precision into %nmmD
cvtss2sd %xmm0O, %xmmO Convert to doubl e precision
movsd %xmmO, 8(%rsp) Store doubl e precision

but, for some reasoigcc prefers the stack-based version.

9 Conclusions

Both AMD and the authors agccdeserve credit for moving x86 processors into a new era. dinetflation

of both the x86-64 hardware and the programming conventtbasged the processor from one that relied
heavily on the stack to hold program state to one where thé neawily used part of the state is held in the
much faster and expanded register set. Finally, x86 hashtaygto ideas developed for RISC processors
in the early 1980s!

x86-64 processors have become the widespread choice foehid machines, such as for database servers
and scientific computing. In the marketplace, they have tivamtage is that they can run 32-bit code with

performance equal to the fastest 32-bit x86 machines. Tdgebt drawback in going from 32-bit applica-
tions to 64 is that the pointer variables double in size, amcksmany data structures contain pointers, this
means that the overall memory requirement can nearly dotible transition from 32 to 64-bit applications
has only occurred for ones having memory needs that excedoh gigabyte address space limitation of
IA32. History has shown that applications grow to use alllatiée processing power and memory size, and
so we can reliably predict that, within the decade, 64-litpssors will be commonplace, even for ordinary
desktop and laptop machines..

Practice Problem Solutions

Problem 1 Solution: [Pg. 9]

This problem illustrates some of the subtleties of type eosion and the different move instructions. In
some cases, we make property thatrtiwyl instruction will set the upper 32 bits of the destinationistsy
to 0's. Some of the problems have multiple solutions.

Tk Ty Instruction | S D Explanation

long long movq %rdi | %rax | No conversion

int long movslq %edi | %rax | Sign extend

char long movshq %dil | %rax | Sign extend
unsigned int unsigned long movl %edi | %eax | Zero extends to 64 bits
unsigned char unsigned long movzhq %dil | %rax | Zero extend to 64
unsigned char unsigned long movzbl %dil | %eax | Zero extends to 64 bits

long int movslq %edi | %rax | Sign extends to 64 bits

long int movl %edi | %eax | Zero extends to 64 bits
unsigned long unsigned movl %edi | %eax | Zero extends to 64 bits

We show that théong to int conversion can use eithamovslq or movl, even though one will sign
extend the upper 32 bits, while the other will zero extendlitis is because the values of the upper 32 bits
are ignored for any subsequent instruction haiteax as an operand.

Problem 2 Solution: [Pg. 11]
We can step through the code fmithprob ~ and determine the following:

1. Themovslq instruction sign extendd to a long integer. This implies thdthas typent andc has
typelong .

2. Themovsbl instruction sign extends to an integer. This means thlathas typechar anda has
typeint .

3. The sum is computed usindeal instruction, indicating that the return value has typte .
From this, we can determine that the unique prototypafidhprob s
int arithprob(int a, char b, long c, int d);
Problem 3 Solution: [Pg. 15]
This problem provides a chance to study the use of condlitinoaes.
A. The operator is/”’. We see this is an example of dividing by a power of two by tighifting (see

CS:APP Section 2.3.7). Before shifting by= 2, we must add a bias @f — 1 = 3 when the dividend
is negative.

B. Here is an annotated version of the assembly code.

x86-64 inplenmentation of arith
X in register %di

1 arith:

2 leal 3(%rdi), %eax X+3

3 cmpl $-1, %edi Conpare x:-1

4 cmovle %eax, %edi If <= then x = x+3
5 sarl $2, %edi X >>= 2

6 movl %edi, %eax return val ue in %ax
7 ret

Thecmovle instruction conditionally changes the number to be shifitesh x to x+3 whenx < —1.

Problem 4 Solution: [Pg. 19]

Reverse engineering code is a good way to understand maelrgigorogramming. This example gives us
a chance to look at the new loop structure.

The following is an annotated version of the assembly code

x86-64 inpl ementation of puzzle
ainregister %di, b in register %esi

1 puzzle:

i isinregister %si, whichis initialized to a
2 movslq  %esi,%rdx result = (long) b
3 jmp .L60 goto middle
4 L61: | oop:
5 movslqg  %edi,%rax (long) i
6 subl %esi, %edi i -=b
7 imulq %rax, %rdx result == (long) i
8 .L60: mi ddl e:
9 testl %edi, %edi Test i
10 g .L61 If >0 goto |oop
11 movq %rdx, %Yrax Return result
12 ret
A. Variableresult isin register%rax.

B. Variablei is in registebesi. Since this is an argument registens initialized toa.

C. Here are the definitions of the five expressions

#define EXPR1 b
#define EXPR2 a
#define EXPR3 0
#define EXPR4 b
#define EXPR5 i

Problem 5 Solution: [Pg. 22]
We can step through the code facrprob  and determine the following:

1. Theaddl instruction fetches a 32-bit integer from the location giley the third argument register
and adds it to the 32-bit version of the first argument regidteom this, we can infer thdt is the
third argument and is the first argument. In additio, must be a signed or unsigned integer, and
must be a pointer to a signed or unsigned integer.

2. Themovslq instruction sign extends the sum (a copy+tf) to a long integer. From this, we can
infer thatt must be a pointer to a signed integer.

3. Theaddq instruction adds the sign-extended value of the previous teuthe location indicated by
the second argument register. From this, we can inferghatthe second argument and that it is a
pointer to a long integer.

There are two valid prototypes famcrprob , which we namencrprob  _s andincrprob _u. They
differ only in whetherx is signed or unsigned.

void incrprob_s(int x, long x(Q, int  *t);
void incrprob_u(unsigned x, long x(Q, int  *t);

Problem 6 Solution: [Pg. 27]

This function is an example of a leaf function that requiesal storage. It can use space beyond the stack
pointer for its local storage, never altering the stack fin

Stack pointer

$rsp >
Unused]
al3] 16
alz] —24
alll -32
afo] _40
A.
x86-64 inplenmentation of |ocal_array
Argunent i in %edi
1 local_array:
2 andl $3, %edi idx =i &3;
3 movq $2, -40(%rsp) a[0] =2
4 movq $3, -32(%rsp) a[1] =3
5  movq $5, -24(%rsp) a[2] =5
6  movq $7, -16(%rsp) a[3] =7

7 mov(q -40(%rsp,%rdi,8), %rax return a[idx]
ret

C. The function never changes the stack pointer. It stotex &b local values in the region beyond the
stack pointer.

Problem 8 Solution: [Pg. 29]
This problem is similar to CS:APP Practice Problem 3.23 umatated for x86-64.

A. struct P1 { int i; char c; long int j; char d; };

i ¢c j d Total | Alignment
0 4 8 16| 24 8

B. struct P2 { long i; char c; char d; int j; };

i c d | Total | Alignment
0 8 9 12| 16 8

C. struct P3 { short w[3]; char c[3] };

w ¢ | Total | Alignment

0 6| 10 2

D. struct P4 { short w[3]; char *C[3] }
w ¢ | Total | Alignment
0 8| 32 8

E. struct P3 { struct P1 a[2]; struct P2 *p )

a p | Total | Alignment
0 48| 56 8

Problem 7 Solution: [Pg. 28]

A. Register%rbx is used to hold the parameter

B. Since%rbx is callee saved, it must be stored on the stack. Since thigisrily use of the stack for
this function, the code uses push and pop instructions & &g restore the register.

C. x86-64 inplenmentation of recursive factorial function rfact
Argument x in % di
1 rfact:
2 testq  %rdi, %rdi Test x
3 pushq %rhbx Save % bx (callee save)

4 movl $1, %eax result =1

5 mov(q %rdi, %rbx Copy x to % bx

6 jle .L9 if x<=0 goto done
7 leaq -1(%rdi), %rdi xml = x-1

8 call rfact rfact (xmil)

9 imulqg %rbx, %rax result = x*rfact(x)
10 .L9: done:

11 popq %rbx Rest ore % bx

12 ret Ret urn

D. Instead of explicitly decrementing and incrementing skeck pointer, the code can yseshq and
popq to both modify the stack pointer and to save and restoretezgitate.

Problem 9 Solution: [Pg. 32]
These cases can be handled by selecting the appropriagefrentrthe table in Figure 10.

Tx Ty Instruction S D
long double | cvtsi2sdq %rdi %xmm0O
double int cvitsd2si %xmmO %eax
float double | cvtss2sd %xmm0O %xmm0O
long float cvtsi2ssq %rdi %xmm0O
float long cvtss2siq %xmm0O %rax

Problem 10 Solution: [Pg. 34]

This problem involves the same reasoning as was requirezktthat numbers declared at label encode 1.8,
but with a simpler example.

We see that the two values are 0 and®(Q)). From the high-order bytes, we can extract an exponent field
of 0x0 (0), from which we subtract a bias of 1023 to get an exponeni@?3. Concatenating the fraction
bits of the two values, we get a fraction field of 0, but with iimplied leading value giving value 1.0. The
constant is therefore.0 x 271923 = 32.0.

| ndex

NaN floating-point not-a-number, 34 %r14w [x86-64] Low order 16 bits of registéorl4, 7
GAS, Gnu assembler, 3 %r14 [x86-64] Low order 8 of registe¥orl4, 7

GAS, Gnu assembler, argument listing, 34 %r14 [x86-64] program register, 7

%xmmpReturn floating-point value register, 31 %r15d [x86-64] Low order 32 bits of registéorl5, 7
%xmmg@return floating-point value, 36 %r15w [x86-64] Low order 16 bits of registéorl5, 7
%ah[x86-64] Bits 8—15 of registe¥orax, 7 %r15 [x86-64] Low order 8 of registe¥orl5, 7

%al [x86-64] Low order 8 of registe¥orax, 7 %r15 [x86-64] program register, 7

%ax[x86-64] Low order 16 bits of registé&orax, 7 %r8d [x86-64] Low order 32 bits of regist&b6r8, 7
%bh[x86-64] Bits 8—15 of registe¥orbx, 7 %r8w [x86-64] Low order 16 bits of regist&b6r8, 7
%bl [x86-64] Low order 8 of registeborbx, 7 %r8 [x86-64] Low order 8 of registe¥or8, 7

%bpl [x86-64] Low order 8 of registe¥orbp, 7 %r8 [x86-64] program register, 7

%bp[x86-64] Low order 16 bits of registé&orbp, 7 %r9d [x86-64] Low order 32 bits of registé&b6r9, 7
%bx [x86-64] Low order 16 bits of registé&orbx, 7 %r9w [x86-64] Low order 16 bits of registé&b6r9, 7
%ch [x86-64] Bits 8—15 of registe¥orcx , 7 %r9 [x86-64] Low order 8 of registe¥or9, 7

%ocl [x86-64] Low order 8 of registeborex , 7 %r9 [x86-64] program register, 7

%cx [x86-64] Low order 16 bits of registérex , 7 %rax [x86-64] program register, 7

%dh[x86-64] Bits 8—15 of registe¥ordx, 7 %rbp [x86-64] program register, 7

%dil [x86-64] Low order 8 of registe¥ordi , 7 %rbx [x86-64] program register, 7

%di [x86-64] Low order 16 bits of registé&ordi , 7 %rcx [x86-64] program register, 7

%dlI [x86-64] Low order 8 of registebordx, 7 %rdi [x86-64] program register, 7

%dx [x86-64] Low order 16 bits of registé&ordx, 7 %rdx [x86-64] program register, 7

%eax [x86-64] Low order 32 bits of registébrax, 7 %rip [x86-64] program counter, 8

%ebp[x86-64] Low order 32 bits of registébrbp, 7 %rsi [x86-64] program register, 7

%ebx [x86-64] Low order 32 bits of registébrbx, 7 %rsp [x86-64] Stack pointer register, 23

%ecx [x86-64] Low order 32 bits of registéforex , 7 %rsp [x86-64] stack pointer register, 7

%edi [x86-64] Low order 32 bits of registébrdi , 7 %sil [x86-64] Low order 8 of registedorsi , 7

%edx [x86-64] Low order 32 bits of registébrdx, 7 %si [x86-64] Low order 16 bits of registé&orsi , 7
%esi [x86-64] Low order 32 bits of registéfrsi , 7 %spl [x86-64]Low order 8 of stack pointer registrsp,

%esp [x86-64] Low order 32 bits of stack pointer register 7
%rsp, 7 %sp [x86-64] Low order 16 bits of stack pointer register
%r10d [x86-64] Low order 32 bits of registéorl0, 7 %rsp, 7
%r10w [x86-64] Low order 16 bits of registéorl0, 7 128-bit arithmetic inccc, 12
%r10 [x86-64] Low order 8 of registe¥or10, 7 8086, Intel 16-bit microprocessor, 1, 30
%r10 [x86-64] program register, 7 8087, floating-point coprocessor, 3, 30

%r11d [x86-64] Low order 32 bits of registéorll, 7

%r11w [x86-64] Low order 16 bits of registéril, 7 ABI (Application Binary Interface), 3
%r11 [x86-64] Low order 8 of registe¥or1l, 7 addq [x86-64] add quad word, 10
%r11 [x86-64] program register, 7 alignment, with x86-64, 29

%r12d [x86-64] Low order 32 bits of registéri2, 7 andq [x86-64] and quad word, 10
%r12w [x86-64] Low order 16 bits of registénri2, 7 Application Binary Interface, 3
%r12 [x86-64] Low order 8 of registe¥orl2, 7 )

%r12 [x86-64] program register, 7 callee (procedure being called), 23
%r13d [x86-64] Low order 32 bits of register13, 7 callee save, 25 _

%r13w [x86-64] Low order 16 bits of register13, 7 caller (procedure making call), 23
%r13 [x86-64] Low order 8 of registe¥6r13, 7 caller save, 25

%r13 [x86-64] program register, 7 cltg [x86-64] convert double word to quad word, 12

%r14d [x86-64] Low order 32 bits of registerr14, 7 ~ cmova [x86-64] move if unsigned greater, 15

cmovae [x86-64] move if unsigned greater or equal, 15mulq [x86-64] signed multiply, 12

cmovb [x86-64] move if unsigned less, 15 incq [x86-64]increment quad word, 10
cmovbe [x86-64] move if unsigned less or equal, 15 .
cmovg [x86-64] move if greater, 15 jp [x86-64] Jump when parity flag set, 35

cmovge [x86-64] move if greater or equal, 15

cmovl [x86-64] move if less, 15

cmovle [x86-64] move if less or equal, 15

cmovna [x86-64] move if not unsigned greater, 15

cmovnae [x86-64] move if unsigned greater or equal, 1

cmovnb [x86-64] move if not unsigned less, 15

cmovnbe [x86-64] move if not unsigned less or equar,n
15

cmovne [x86-64] move if not equal, 15

cmovng [x86-64] move if not greater, 15

cmovnge [x86-64] move if not greater or equal, 15

cmovnl [x86-64] move if not less, 15

cmovnle [x86-64] move if not less or equal, 15

cmovns [x86-64] move if nonnegative, 15 movss [x86-64] Move single precision, 30

cmovnz [x86-64] move if not zero, 15
. : movzbq [x86-64] move zero-extend byte to quad word,
cmovs [x86-64] move if negative, 15 a [9 | y a

cmovz [x86-64] move if zero, 15
cmpq [x86-64] compare quad words, 13

cqto [x86-64] convert quad word to oct word, 12 notq [x86-64] complement quad word, 10
cvtsd2ss  [x86-64] Convert single to double precision,

30 orq [x86-64]or quad word, 10
cvtsi2sd  [x86-64] Convertinteger to double precision,

30 PC (program counter) relative, 8

cvtsi2sdq  [x86-64] Convert quadword integer to douPopd [x86-64] pop quad word, 9
ble precision, 30 pushq [x86-64] push quad word, 9

cvtsi2ss  [x86-64] Convert integer to single precision ) )
30 Red zone, region beyond stack pointer, 27

Reduced-Instruction Set Computers (RISC), 1, 37
gle precision, 30 rep [IA32] String repeat instruction, used as no-op, 18

cvtss2sd  [x86-64] Convert single to double precision’€t [x86-64] Return from procedure call, 18
30 RISC, Reduced-Instruction Set Computers, 1, 37

leaf procedure, 22
leag [x86-64]load effective address, 10

MX, Intel multimedia extenstion, 29

ovabsq [x86-64] move absolute quad word, 9

ovaps [x86-64] used to copy values between XMM
registers, 37

movlpd [x86-64] similar tomovsd, 31, 37

movq [x86-64] move quad word, 9

movsbq [x86-64] move sign-extend byte to quad word, 9

movsd [x86-64] Move double precision, 30

movslq [x86-64] move sign-extend double word to quad

word, 9

mulg [x86-64] unsigned multiply, 12

cvtsi2ssq  [x86-64] Convert quadword integer to sin

cvttsd25|ger [>3<g6-64] Convert double precision to |nte-Salq [x86-64] shift left quad word, 10

cvitsd2siq  [x86-64] Convert double precision to qua ?Irc? Kgggﬂ zﬂ:g @;ﬁg?:g%é?gt ggad word, 10
cvttssZsiWor?ggée-gj]r'ggnvert single precision to inte-Shrq [X$6_64] shift Iogical right double word, Z_LO
er 30 SIMD, Sl_ngle-lnstrgctlon Multiple Data execution, 29
cvttss23i(§J ' [x86-64] Convert single precision to quad§peculat|ve ex_ecutlon, 14 . .
; SSES3, Streaming SIMD Extensions, version 3, 29

word integer, 30

subq [x86-64] subtract quad word, 10

decq [x86-64] decrement quad word, 10

divgq [x86-64] unsigned divide, 12 testq  [x86-64]test quad word, 13

ucomisd [x86-64] compare double precision, 34
ucomiss [x86-64] compare single precision, 34
Unordered, floating-point comparision outcome, 35

idivg [x86-64] signed divide, 12
imulg [x86-64] multiply quad word, 10

x86, Intel instruction set, 1

x87, floating-point instructions for 8087, 30

XMM, SSE3 registers, 29

xorps [x86-64] used to clear XMM register, 31, 37
xorg [x86-64] exclusive-or quad word, 10

